

# FOODiE, Fortran Object oriented Ordinary Differential Equations integration library based on Abstract Calculus Pattern

Zaghi S.<sup>a,1,\*</sup>, Curcic M.<sup>b,2</sup>, Rouson D.<sup>c,3</sup>, Beekman I.<sup>c,4</sup>

<sup>a</sup>CNR-INSEAN, Istituto Nazionale per Studi ed Esperienze di Architettura Navale, Via di Vallerano 139, Rome, Italy, 00128

<sup>b</sup>Ocean Sciences Rosenstiel School of Marine and Atmospheric Science, University of Miami, 4600 Rickenbacker Causeway Miami, FL  
33149-1098 +1 305.421.4000

<sup>c</sup>Sourcery Institute 482 Michigan Ave., Berkeley, CA 94707

---

## Abstract

To be written.

**Keywords:** Ordinary Differential Equations (ODE), Partial Differential Equations (PDE), Object Oriented Programming (OOP), Abstract Calculus Pattern (ACP), Fortran,

---

## PROGRAM SUMMARY

*Manuscript Title:* FOODiE, Fortran Object oriented Ordinary Differential Equations integration library based on Abstract Calculus Pattern

*Authors:* Zaghi, S., Curcic, M., Rouson, D., Beekman, I.

*Program title:* FOODiE

*Journal Reference:*

*Catalogue identifier:*

*Licensing provisions:* GNU General Public License (GPL) v3

*Programming language:* Fortran (standard 2008 or newer); developed and tested with GNU gfortran 5.2 or newer

*Computer(s) for which the program has been designed:* designed for shared-memory multi-cores workstations and for hybrid distributed/shared-memory supercomputers, but any computer system with a Fortran (2008+) compiler is suited

*Operating system(s) for which the program has been designed:* designed for POSIX architecture and tested on GNU/Linux one

*RAM required to execute with typical data: bytes:* [1MB, 1GB] × core, simulation-dependent

*Has the code been vectorised or parallelized?:* the library is not aware of the parallel back-end, it providing a high-level models, but the provided tests suite shows parallel usage by means of MPI library and OpenMP paradigm

*Number of processors used:* tested up to 256

*Supplementary material:*

*Keywords:* ODE, PDE, OOP, ACP, Fortran

*CPC Library Classification:* 4.3 Differential Equations, 4.10 Interpolation, 12 Gases and Fluids

*External routines/libraries used:*

*CPC Program Library subprograms used:*

*Nature of problem:*

Numerical integration of (general) Ordinary Differential Equations system

*Solution method:*

---

\*Corresponding author

Email addresses: stefano.zaghi@cnr.it (Zaghi S.), milan@orca.rsmas.miami.edu (Curcic M.), damian@sourceryinstitute.org (Rouson D.), izaak@izaakbeekman.com (Beekman I.)

<sup>1</sup>Ph. D., Aerospace Engineer, Research Scientist, Dept. of Computational Hydrodynamics at CNR-INSEAN.

<sup>2</sup>Ph.D. Meteorology and Physical Oceanography, Research Scientist, Dept. of Ocean Sciences Rosenstiel School of Marine and Atmospheric Science at University of Miami

<sup>3</sup>Ph.D. Mechanical Engineering, Founder and President Sourcery Institute and Sourcery, Inc.

<sup>4</sup>Graduate Research Assistant, Princeton/UMD CCROCCO LAB

*Restrictions:*

*Unusual features:*

*Additional comments:*

*Running time:*

*References:*

## 1. Introduction

### 1.1. Background

Initial Value Problem (IVP, or Cauchy problem) constitutes a class of mathematical models of paramount relevance, it being applied to the modelling of a wide range of problems. Briefly, an IVP is an Ordinary Differential Equations system (ODE) coupled with specified initial values of the unknown state variables, the solution of which are searched at a given time after the initial time considered.

The prototype of IVP can be expressed as:

$$\begin{aligned} U_t &= R(t, U) \\ U_0 &= U(t_0) \end{aligned} \tag{1}$$

where  $U(t)$  is the vector of state variables being a function of the time-like independent variable  $t$ ,  $U_t = \frac{dU}{dt} = R(t, U)$  is the (vectorial) residuals function and  $U(t_0)$  is the (vectorial) initial conditions, namely the state variables function evaluated at the initial time  $t_0$ . In general, the residuals function  $R$  is a function of the state variable  $U$  through which it is a function of time, but it can be also a direct function of time, thus in general  $R = R(t, U(t))$  holds.

The problem prototype 1 is ubiquitous in the mathematical modelling of physical problems: essentially whenever an *evolutionary* phenomenon is considered the prevision (simulation) of the future solutions involves the solution of an IVP. As a matter of facts, many physical problems (fluid dynamics, chemistry, biology, evolutionary-anthropology, etc...) are described by means of an IVP.

It is worth noting that the state vector variables  $U$  and its corresponding residuals function  $U_t = \frac{dU}{dt} = R(t, U)$  are *problem dependent*: the number and the meaning of the state variables as well as the equations governing their evolution (that are embedded into the residuals function) are different for the Navier-Stokes conservation laws with respect the Burgers one, as an example. Nevertheless, the *solvers* used for the prediction of the Navier-Stokes equations evolution (hereafter the *solution*) are the same that are used for Burgers equations time-integration. As a consequence, the solution of the IVP model prototype can be generalized, allowing the application of the same solver to many different problems, thus eliminating the necessity to re-implement the same solver for each different problem.

FOODiE library is designed for solving the generalized IVP 1, it being completely unaware of the actual problem's definition. FOODiE library provides a high-level, well-documented, simple Application Program Interface (API) for many well-known ODE integration solvers, its aims being twofold:

- provide a robust set of ODE solvers ready to be applied to a wide range of different problems;
- provide a simple framework for the rapid development of new ODE solvers.

Add citations to IVP, Cauchy, ODE references.

### 1.2. Related Works

To be written.

### 1.3. Motivations and Aims

FOODiE library must:

- be written in modern Fortran (standard 2008 or newer);
- be written by means of Object Oriented Programming (OOP) paradigm;

- be well documented;
- be Tests Driven Developed (TDD);
- be collaboratively developed;
- be free.

FOODiE, meaning Fortran Object oriented Ordinary Differential Equations integration library, has been developed with the aim to satisfy the above specifications. The present paper is its first comprehensive presentation.

The Fortran (Formula Translator) programming language is the *de facto* standard into computer science field: it strongly facilitates the effective and efficient translation of (even complex) mathematical and numerical models into an operative software without compromise on computations speed and accuracy. Moreover, its simple syntax is suitable for scientific researchers that are interested (and skilled) in the physical aspects of the numerical computations rather than computer technicians. Consequently, we develop FOODiE using Fortran language: FOODiE is written by research scientists for research scientists.

One key-point of the FOODiE development is the *problem generalization*: the problem solved must be the IVP 1 rather than any of its actual definitions. Consequently, we must rely on a *generic* implementation of the solvers. To this aim, OOP is very useful: it allows to express IVP 1 in a very concise and clear formulation that is really generic. In particular, our implementation is based on *Abstract Calculus Pattern* (ACP) concept.

*The Abstract Calculus Pattern.* The abstract calculus pattern provides a simple solution for the connection between the very high-level expression of IVP 1 and the eventual concrete (low-level) implementation of the ODE problem being solved. ACP essentially constitutes a *contract* based on an Abstract Data Type (ADT): we specify an ADT supporting a certain set of mathematical operators (differential and integral ones) and implement FOODiE solvers only on the basis of this ADT. FOODiE clients must formulate the ODE problem under integration defining their own concrete extensions of our ADT (implementing all the deferred operators). Such an approach defines the abstract calculus pattern: FOODiE solvers are aware of only the ADT, while FOODiE clients extend the ADT for defining the concrete ODE problem.

Is is worth noting that this ACP emancipates the solvers implementations from any low-level problem-dependent details: the ODE solvers developed with this pattern are extremely concise, clear, maintainable and less errors-prone with respect a low-level (non abstract) pattern. Moreover, the FOODiE clients can use solvers being extremely robust: as a matter of facts, FOODiE solvers are expressed in a formulation very close to the mathematical one and are tested on an extremely varying family of problems. As shown in the following, such a great flexibility does not compromise the computational efficiency.

The present paper is organized as following: in section 2 a brief description of the mathematical and numerical methods currently implemented into FOODiE is presented; in section 3 a detailed discussion on the implementation specifications is provided by means of an analytical code-listings review; in section 4 a verification analysis on the results of FOODiE applications is presented; section 5 provides an analysis of FOODiE performances under parallel frameworks scenario like the OpenMP and MPI paradigms; finally, in section 6 concluding remarks and perspectives are depicted.

Add citations to Fortran standards, OOP, TDD, free software, ACP, ADT, Rouson's book references.

## 2. Mathematical and Numerical Models

In many (most) circumstances, the solution of equation 1 cannot be computed in a closed, exact form (even if it exists and is unique) due to the complexity and nature of the residuals functions, that is often non linear. Consequently, the problem is often solved relying on a numerical approach: the solution of system 1 at a time  $t^n$ , namely  $U(t^n)$ , is approximated by a subsequent time-marching approximations  $U_0 = u_0 \rightarrow u_1 \rightarrow u_2 \rightarrow \dots \rightarrow u_N \approx U(t^n)$  where the relation  $u_i \rightarrow u_{i+1}$  implies a *stepping, numerical integration* from the time  $t^i$  to time  $t^{i+1}$  and  $N$  is the total number of numerical time steps necessary to evolve the initial conditions toward the searched solution  $U(t^n)$ . To this aim, many numerical schemes have been devised. Notably, the numerical schemes of practical usefulness must posses some necessary proprieties such as *consistency* and *stability* to ensure that the numerical approximation *converges*

to the exact solution as the numerical time step tends to zero. A detailed discussion of these details is out the scope of the present work and is omitted. Here, we briefly recall some classifications necessary to introduce the schemes implemented into the FOODiE library.

A non comprehensive classification of the most widely used schemes could distinguish between *multi-step* versus *one-step* schemes and between *explicit* versus *implicit* schemes.

Essentially, the multi-step schemes have been developed to obtain an accurate approximation of the subsequent numerical steps using the informations contained into the previously computed steps, thus this approach relates the next step approximation to a set of the previously computed steps. On the contrary, a one-step scheme evolves the solution toward the next step using only the information coming from the current time approximation. In the framework of one-step schemes family an equivalent accurate approximation can be obtained by means of a multi-stage approach as the one due to Runge-Kutta. The current version of FOODiE provides schemes belonging to both these families.

The other ODE solvers classification concerns with explicit or implicit nature of the schemes employed. Briefly, an explicit scheme computes the next step approximation using the previously computed steps at most to the current time, whereas an implicit scheme uses also the next step approximation (that is the unknown), thus it requires extra computations. The implicit approach is of practical use for *stiff* systems where the usage of explicit schemes could require an extremely small time step to evolve in a *stable* way the solution. Mixing together explicit and implicit schemes it is possible to build a family of *predictor-corrector* methods: using an explicit scheme to predict a guess for the next step approximation it is possible to use an implicit method for correcting this guess. Currently, FOODiE provides explicit solvers and predictor-correct ones.

FOODiE currently implements the following ODE schemes:

- one-step schemes:
  - explicit forward Euler scheme, it being 1<sup>st</sup> order accurate;
  - explicit Runge-Kutta schemes:
    - \* TVD/SSP Runge-Kutta schemes:
      - 2-stages, it being 2<sup>nd</sup> order accurate;
      - 3-stages, it being 3<sup>rd</sup> order accurate;
      - 5-stages, it being 4<sup>th</sup> order accurate;
    - \* low storage Runge-Kutta schemes:
      - 5-stages 2N registers schemes, it being 4<sup>th</sup> order accurate;
      - 6-stages 2N registers schemes, it being 4<sup>th</sup> order accurate;
      - 7-stages 2N registers schemes, it being 4<sup>th</sup> order accurate;
- multi-step schemes:
  - explicit Adams-Bashforth schemes:
    - \* 2-steps, it being 2<sup>nd</sup> order accurate;
    - \* 3-steps, it being 3<sup>rd</sup> order accurate;
    - \* 4-steps, it being 4<sup>th</sup> order accurate;
  - implicit Adams-Moulton schemes:
    - \* 1-step, it being 2<sup>nd</sup> order accurate;
    - \* 2-steps, it being 3<sup>rd</sup> order accurate;
    - \* 3-steps, it being 4<sup>th</sup> order accurate;
  - predictor-corrector Adams-Bashforth-Moulton schemes:
    - \* 1-step, it being 2<sup>nd</sup> order accurate;
    - \* 2-steps, it being 3<sup>rd</sup> order accurate;
    - \* 3-steps, it being 4<sup>th</sup> order accurate;

- explicit Leapfrog schemes<sup>5</sup>:
  - \* 2-steps unfiltered, it being 2<sup>nd</sup> order accurate, but mostly *unstable*;
  - \* 2-steps Robert-Asselin filtered, it being 1<sup>st</sup> order accurate (on *amplitude error*);
  - \* 2-steps Robert-Asselin-Williams filtered, it being 3<sup>rd</sup> order accurate (on *amplitude error*);

Add citations to all solvers reference papers.

### 2.1. The explicit forward Euler scheme

The explicit forward Euler scheme for ODE integration is probably the simplest solver ever devised. Considering the system 1, the solution (approximation) of the state vector  $U$  at the time  $t^{n+1} = t^n + \Delta t$  assuming to known the solution at time  $t^n$  is:

$$U(t^{n+1}) = U(t^n) + \Delta t \cdot R[t^n, U(t^n)] \quad (2)$$

where the solution at the new time step is computed by means of only the current time solution, thus this is an explicit scheme. The solution is an approximation of 1<sup>st</sup> order, the local truncation error being  $O(\Delta t^2)$ . As well known, this scheme has an absolute (linear) stability locus equals to  $|1 + \Delta t \lambda| \leq 1$  where  $\lambda$  contains the eigenvalues of the linear (or linearized) Jacobian matrix of the system.

This scheme is Total Variation Diminishing (TVD), thus satisfies the maximum principle (or the equivalent positivity preserving property).

Add citations.

### 2.2. The explicit TVD/SSP Runge-Kutta class of schemes

Runge-Kutta methods belong to the more general multi-stage family of schemes. This kind of schemes has been designed to achieve a more accurate solution than the 1<sup>st</sup> Euler scheme, but without increasing the number of time steps used, as it is done with the multi-step schemes. Essentially, the high order of accuracy is obtained by means of *intermediate values* (the stages) of the solution and its derivative are generated and used within a single time step. This commonly implies the allocation of some auxiliary memory registers for storing the intermediate stages.

Notably, the multi-stage schemes class has the attractive property to be *self-starting*: the high order accurate solution can be obtained directly from the previous one, without the necessity to compute *before* a certain number of previous steps, as it happens for the multi-step schemes. Moreover, one-step multi-stage methods are suitable for adaptively-varying time-step size (that is also possible for multi-step schemes, but at a cost of more complexity) and for discontinuous solutions, namely discontinued solutions happening at a certain time  $t^*$  (that in a multi-step framework can involve an overall accuracy degradation).

In general, the TVD/SSP Runge-Kutta schemes provided by FOODiE library are written by means of the following algorithm:

$$U^{n+1} = U^n + \Delta t \cdot \sum_{s=1}^{N_s} \beta_s K_s \quad (3)$$

where  $N_s$  is the number of Runge-Kutta stages used and  $K_s$  is the  $s^{th}$  stage defined as:

$$K_s = R\left(t^n + \gamma_s \Delta t, U^n + \Delta t \sum_{l=1}^{s-1} \alpha_{s,l} K_l\right) \quad (4)$$

It is worth noting that the equations 3 and 4 contain also implicit schemes. A scheme belonging to this family is operative once the coefficients  $\alpha, \beta, \gamma$  are provided. We represent these coefficients using the Butcher's table, that for an explicit scheme where  $\gamma_1 = \alpha_{1,*} = \alpha_{i,i} = 0$  has the form reported in table 1.

---

<sup>5</sup>Leapfrog method is a 2-steps solver being, in general 2<sup>nd</sup> order accurate, but mostly *unstable*. It is well suited for periodic/oscillatory problems, thus the error is generally categorized accordingly to the phase and amplitude errors: the unfiltered scheme provides a 2<sup>nd</sup> order error on phase approximation while the amplitude error is null. However, the unfiltered scheme is mostly unstable, thus many filters have been devised. In general, the filters retain the accuracy on the phase error, while affect the amplitude one.

Table 1: Butcher's table for explicit Runge-Kutta schemes

$\gamma_2$	$\alpha_{2,1}$				
$\gamma_3$	$\alpha_{3,1}$	$\alpha_{3,2}$			
$\vdots$	$\vdots$		$\ddots$		
$\gamma_{N_s}$	$\alpha_{N_s,1}$	$\alpha_{N_s,2}$	$\cdots$	$\alpha_{N_s,N_s-1}$	
	$\beta_1$	$\beta_2$	$\cdots$	$\beta_{N_s-1}$	$\beta_{N_s}$

The equations 3 and 4 show that Runge-Kutta methods do not require any additional differentiations of the ODE system for achieving high order accuracy, rather they require additional evaluations of the residuals function  $R$ .

The nature of the scheme and the properties of the solutions obtained depend on the number of stages and on the value of the coefficients selected. Currently, FOODiE provides 3 Runge-Kutta schemes having TVD or Strong Stability Preserving (SSP) propriety (thus they being suitable for ODE systems involving rapidly changing non linear dynamics) the Butcher's coefficients of which are reported in tables 2, 3, 4.

Table 2: Butcher's table of 2 stages, 2<sup>nd</sup> order, Runge-Kutta TVD scheme

1	1	0	
	1/2	1/2	

Table 3: Butcher's table of 3 stages, 3<sup>rd</sup> order, Runge-Kutta SSP scheme

1	1		
1/2	1/4	1/4	
	1/6	1/6	2/3

Table 4: Butcher's table of 5 stages, 4<sup>th</sup> order accurate, Runge-Kutta SSP scheme

0.39175222700392	0.39175222700392				
0.58607968896779	0.21766909633821	0.36841059262959			
0.47454236302687	0.08269208670950	0.13995850206999	0.25189177424738		
0.93501063100924	0.06796628370320	0.11503469844438	0.20703489864929	0.54497475021237	
	0.14681187618661	0.24848290924556	0.10425883036650	0.27443890091960	0.22600748319395

The absolute stability locus depends on the coefficients selected, however, as a general principle, we can assume that greater is the stages number and wider is the stability locus on equal accuracy orders.

It is worth noting that FODDiE also provides a one-stage TVD Runge-Kutta solver that reverts back to the explicit forward Euler scheme: it can be used, for example, into a Recursive Order Reduction (ROR) framework that automatically checks some properties of the solution and, in case, reduces the order of the Runge-Kutta solver until those properties are obtained.

Add citations.

### 2.3. The explicit low storage Runge-Kutta class of schemes

As aforementioned, standard Runge-Kutta schemes have the drawback to require  $N_s$  auxiliary memory registers to store the necessary stages data. In order to make an efficient use of the available limited computer memory, the class of low storage Runge-Kutta scheme was devised. Essentially, the standard Runge-Kutta class (under some specific conditions) can be reformulated allowing a more efficient memory management. Currently FOODiE provides a class of  $2N$  registers storage Runge-Kutta schemes, meaning that the storage of all stages requires only 2 registers of memory with a *word* length  $N$  (namely the length of the state vector) in contrast to the standard formulation where  $N_s$  registers of the same length  $N$  are required. This is a dramatic improvement of memory efficiency especially for

schemes using a high number of stages ( $N_s \geq 4$ ) where the memory necessary is an half with respect the original formulation. Unfortunately, not all standard Runge-Kutta schemes can be reformulated as a low storage one.

Following the Williamson's approach the standard coefficients are reformulated to the coefficients vectors  $A$ ,  $B$  and  $C$  and the Runge-Kutta algorithm becomes:

$$\begin{aligned} K_1 &= U(t^n) \\ K_2 &= 0 \\ K_2 = A_s K_2 + \Delta t \cdot R(t^n + C_s \Delta t, K_1) \\ K_1 = K_1 + B_s K_2 \\ U(t^{n+1}) &= K_1 \end{aligned} \quad s = 1, 2, \dots, N_s \quad (5)$$

Currently FOODiE provides 5/6/7 stages, all 4<sup>th</sup> order, 2N registers explicit schemes, the coefficients of which are listed in tables 5, 6 and 7.

Table 5: Williamson's table of 5 stages, 4<sup>th</sup> order, Runge-Kutta low storage scheme

Stage	$A$	$B$	$C$
1	0	$\frac{1432997174477}{1432997174477}$	0
2	$-\frac{567301805773}{1357537059087}$	$\frac{9575080441755}{3161836677171}$	$\frac{1432997174477}{9575080441755}$
3	$-\frac{2404267990393}{2016746695238}$	$\frac{13612068292357}{1720146321549}$	$\frac{9575080441755}{2526289341429}$
4	$-\frac{355091886646}{2091501179385}$	$\frac{2090206949498}{314564353532}$	$\frac{6820363962896}{200634551931}$
5	$-\frac{1275896237668}{842570457699}$	$\frac{4481467310338}{14882151754819}$	$\frac{3224310063776}{2802321613138}$

Table 6: Williamson's table of 6 stages, 4<sup>th</sup> order, Runge-Kutta low storage scheme

Stage	$A$	$B$	$C$
1	0	0.122000000000	0
2	-0.691750960670	0.477263056358	0.122000000000
3	-1.727127405211	0.381941220320	0.269115878630
4	-0.694890150986	0.447757195744	0.447717183551
5	-1.039942756197	0.498614246822	0.749979795490
6	-1.531977447611	0.186648570846	0.898555413085

Table 7: Williamson's table of 7 stages, 4<sup>th</sup> order, Runge-Kutta low storage scheme

Stage	$A$	$B$	$C$
1	0.000000000000	0.117322146869	0.000000000000
2	-0.647900745934	0.503270262127	0.117322146869
3	-2.704760863204	0.233663281658	0.294523230758
4	-0.460080550118	0.283419634625	0.305658622131
5	-0.500581787785	0.540367414023	0.582864148403
6	-1.906532255913	0.371499414620	0.858664273599
7	-1.450000000000	0.136670099385	0.868664273599

Similarly to the TVD/SSP Runge-Kutta class, the low storage class also provides a fail-safe one-stage solver reverting back to the explicit forward Euler solver, that is useful for ROR-like frameworks.

Add citations.

#### 2.4. The explicit Adams-Bashforth class of schemes

Adams-Bashforth methods belong to the more general (linear) explicit multi-step family of schemes. This kind of schemes has been designed to achieve a more accurate solution than the 1<sup>st</sup> Euler scheme using the information coming from the solutions already computed at previous time steps. Typically only one new residuals function  $R$  evaluation is required at each time step, whereas Runge-Kutta schemes require many of them.

In general, the Adams-Bashforth schemes provided by FOODiE library are written by means of the following algorithm (for only explicit schemes):

$$U(t^{n+N_s}) = U(t^{n+N_s-1}) + \Delta t \sum_{s=1}^{n+N_s} b_s \cdot R[t^{n+s-1}, U(t^{n+s-1})] \quad (6)$$

where  $N_s$  is the number of time steps considered and  $b_s$  are the linear coefficients selected.

Currently FOODiE provides 2, 3, and 4 steps schemes having 2<sup>nd</sup>, 3<sup>rd</sup> and 4<sup>th</sup> formal order of accuracy, respectively. The  $b_s$  coefficients are reported in table 8.

Table 8: Explicit Adams-Bashforth coefficients

$N_s$	$b_1$	$b_2$	$b_3$	$b_4$
2	$-\frac{1}{2}$	$\frac{3}{2}$	/	/
3	$\frac{5}{12}$	$-\frac{16}{12}$	$\frac{23}{12}$	/
4	$-\frac{9}{24}$	$\frac{37}{24}$	$-\frac{59}{24}$	$\frac{55}{24}$

Similarly to the Runge-Kutta classes, the Adams-Bashforth class also provides a fail-safe one-step solver reverting back to the explicit forward Euler solver, that is useful for ROR-like frameworks.

It is worth noting that for  $N_s > 1$  the Adams-Bashforth class of solvers is not *self-starting*: the values of  $U(t^1)$ ,  $U(t^2)$ , ...,  $U(t^{N_s-1})$  must be provided. To this aim, a lower order multi-step scheme or an equivalent order one-step multi-stage scheme can be used.

Add citations.

## 2.5. The implicit Adams-Moulton class of schemes

Adams-Moulton methods belong to the more general (linear) implicit multi-step family of schemes. This kind of schemes has been designed to achieve a more accurate solution than the 1<sup>st</sup> Euler scheme using the information coming from the solutions already computed at previous time steps. Typically only one new residuals function  $R$  evaluation is required at each time step, whereas Runge-Kutta schemes require many of them.

In general, the Adams-Moulton schemes provided by FOODiE library are written by means of the following algorithm (for only implicit schemes):

$$U(t^{n+N_s}) = U(t^{n+N_s-1}) + \Delta t \sum_{s=0}^{n+N_s-1} b_s \cdot R[t^{n+s}, U(t^{n+s})] + b_{N_s} \cdot R[t^{n+N_s}, U(t^{n+N_s})] \quad (7)$$

where  $N_s$  is the number of time steps considered and  $b_s$  are the linear coefficients selected.

Currently FOODiE provides 1, 2, and 3 steps schemes having 2<sup>nd</sup>, 3<sup>rd</sup> and 4<sup>th</sup> formal order of accuracy, respectively. The  $b_s$  coefficients are reported in table 9.

Table 9: Implicit Adams-Moulton coefficients

$N_s$	$b_0$	$b_1$	$b_2$	$b_3$
1	$\frac{1}{2}$	$\frac{1}{2}$	/	/
2	$-\frac{1}{12}$	$\frac{8}{12}$	$\frac{5}{12}$	/
3	$\frac{1}{24}$	$-\frac{5}{24}$	$\frac{19}{24}$	$\frac{9}{24}$

Similarly to the Runge-Kutta and Adams-Bashforth classes, the Adams-Moulton class also provides a fail-safe zero-step solver reverting back to the implicit backward Euler solver, that is useful for ROR-like frameworks.

It is worth noting that for  $N_s > 1$  the Adams-Moulton class of solvers is not *self-starting*: the values of  $U(t^1)$ ,  $U(t^2)$ , ...,  $U(t^{N_s-1})$  must be provided. To this aim, a lower order multi-step scheme or an equivalent order one-step multi-stage scheme can be used.

Add citations.

## 2.6. The predictor-corrector Adams-Bashforth-Moulton class of schemes

Adams-Bashforth-Moulton methods belong to the more general (linear) predictor-corrector multi-step family of schemes. This kind of schemes has been designed to achieve a more accurate solution than the 1<sup>st</sup> Euler scheme using

the information coming from the solutions already computed at previous time steps. Typically only one new residuals function  $R$  evaluation is required at each time step, whereas Runge-Kutta schemes require many of them.

In general, the Adams-Bashforth-Moulton schemes provided by FOODiE library are written by means of the following algorithm:

$$\begin{aligned} U(t^{n+N_s^p})_p &= U(t^{n+N_s^p-1}) + \Delta t \sum_{s=1}^{n+N_s^p} b_s^p \cdot R[t^{n+s-1}, U(t^{n+s-1})] \\ U(t^{n+N_s^c})_c &= U(t^{n+N_s^c-1}) + \Delta t \sum_{s=0}^{n+N_s^c-1} b_s^c \cdot R[t^{n+s}, U(t^{n+s})] + b_{N_s^c}^c \cdot R[t^{n+N_s^p}, U(t^{n+N_s^p})_p] \end{aligned} \quad (8)$$

where  $N_s^{p,c}$  is the number of time steps considered for the Adams-Bashforth predictor/Adams-Moulton corrector (respectively) and  $b_s^{p,c}$  are the corresponding linear coefficients selected. Essentially, the Adams-Bashforth prediction  $U(t^{n+N_s^p})_p$  is corrected by means of the Adams-Moulton correction resulting in  $U(t^{n+N_s^c})_c$ . In order to preserve the formal order of accuracy the relation  $N_s^p = N_s^c + 1$  always holds.

Currently FOODiE provides  $N_s^c = 1, 2, 3 \rightarrow N_s^c = 2, 3, 4$  steps schemes having 2<sup>nd</sup>, 3<sup>rd</sup> and 4<sup>th</sup> formal order of accuracy, respectively. The  $b_s^{p,c}$  coefficients are those reported in tables 8 and 9.

[Add citations.](#)

## 2.7. The leapfrog solver

The *leapfrog* scheme belongs to the multi-step family, it being formally a centered second order approximation in time. The leapfrog method (in its original formulation) is mostly unstable, however it is well suited for periodic-oscillatory problems providing a null error on the amplitude value and a formal second order error on the phase one, under the satisfaction of the time-step size stable limit. Commonly, the leapfrog methods are said to provide a  $2\Delta t$  computational mode that can generate unphysical, unstable solutions. As consequence, the original leapfrog scheme is generally *filtered* in order to suppress these computational modes.

The unfiltered leapfrog scheme provided by FOODiE is:

$$U(t^{n+2}) = U(t^n) + 2\Delta t \cdot R[t^{n+1}, U(t^{n+1})] \quad (9)$$

FOODiE provides, in a *seamless* API, also filtered leapfrog schemes. A widely used filter is due to Robert and Asselin, that suppress the computational modes at the cost of accuracy reduction resulting into a 1<sup>st</sup> order error in amplitude value. A more accurate filter, able to provide a 3<sup>rd</sup> order error on amplitude, is a modification of the Robert-Asselin filter due to Williams known as Robert-Asselin-Williams (RAW) filter, that filters the approximation of  $U(t^{n+1})$  and  $U(t^{n+2})$  by the following scalar coefficient:

$$\begin{aligned} U(t^{n+1}) &= U(t^{n+1}) + \Delta * \alpha \\ U(t^{n+2}) &= U(t^{n+2}) + \Delta * (\alpha - 1) \\ \text{where} \\ \Delta &= \frac{\nu}{2}(U^n - 2U^{n+1} + U^{n+2}) \end{aligned} \quad (10)$$

The filter coefficients should be taken as  $\nu \in (0, 1]$  and  $\alpha \in (0.5, 1]$ . If  $\alpha = 0.5$  the filters of time  $t^{n+1}$  and  $t^{n+2}$  have the same amplitude and opposite sign thus allowing to the optimal 3<sup>rd</sup> order error on amplitude. The default values of the FOODiE provided scheme are  $\nu = 0.01$   $\alpha = 0.53$ , but they can be customized at runtime.

[Add citations.](#)

### 3. Application Program Interface

In this section we review the FOODiE API providing a detailed discussion of the implementation choices.

As aforementioned, the programming language used is the Fortran 2008 standard, that is a minor revision of the previous Fortran 2003 standard. Such a new Fortran idioms provide (among other useful features) an almost complete support for OOP, in particular for ADT concept. Fortran 2003 has introduced the *abstract derived type*: it is a derived type suitable to serve as *contract* for concrete type-extensions that has not any actual implementations, rather it provides a well-defined set of type bound procedures interfaces, that in Fortran nomenclature are called *deferred* procedures. Using such an abstract definition, we can implement algorithms operating on only this abstract type and on *all its concrete extensions*. This is the key feature of FOODiE library: all the above described ODE solvers are implemented on the knowledge of *only one abstract type*, allowing an implementation-style based on a very high-level syntax. In the meanwhile, client codes must implement their own IVPs extending only one simple abstract type.

In the subsection 3.1 a review of the FOODiE main ADT, the *integrand* type, is provided, while subsections 3.2, 3.3, 3.4, 3.5 and 3.8 cover the API of the currently implemented solvers.

It is worth noting that all FOODiE public *entities* (ADT and solvers) must be accessed by the FOODiE module, see listing 1 for an example on how access to all public FOODiE entities.

```
use foodie, only: integrand, &
    adams_bashforth_integrator, &
    adams_moulton_integrator, &
    adams_bashforth_moulton_integrator, &
    euler_explicit_integrator, &
    leapfrog_integrator, &
    ls_runge_kutta_integrator, &
    tvd_runge_kutta_integrator
! or simply
use foodie
```

Listing 1: usage example importing all public entities of FOODiE main module

#### 3.1. The main FOODiE Abstract Data Type: the *integrand* type

The implemented ACP is based on one main ADT, the *integrand* type, the definition of which is shown in listing 2.

```
type, abstract :: integrand
!< Abstract type for building FOODiE ODE integrators.
contains
    ! public deferred procedures that concrete integrand-field must implement
    procedure(time_derivative),      pass(self), deferred, public:: t
    ! operators
    procedure(symmetric_operator),   pass(lhs), deferred, public:: integrand_multiply_integrand
    procedure(integrand_op_real),    pass(lhs), deferred, public:: integrand_multiply_real
    procedure(real_op_integrand),    pass(rhs), deferred, public:: real_multiply_integrand
    procedure(symmetric_operator),   pass(lhs), deferred, public:: add
    procedure(symmetric_operator),   pass(lhs), deferred, public:: sub
    procedure(assignment_integrand), pass(lhs), deferred, public:: assign_integrand
    ! operators overloading
    generic, public:: operator(+) => add
    generic, public:: operator(-) => sub
    generic, public:: operator(*) => integrand_multiply_integrand, &
                                real_multiply_integrand, &
                                integrand_multiply_real
    generic, public:: assignment(=) => assign_integrand
endtype integrand
```

Listing 2: integrand type definition

The *integrand* type does not implement any actual integrand field, it being and abstract type. It only specifies which deferred procedures are necessary for implementing an actual concrete integrand type that can use a FOODiE solver.

As shown in listing 2, the number of the deferred type bound procedures that clients must implement into their own concrete extension of the *integrand* ADT is very limited: essentially, there are 1 ODE-specific procedure plus some operators definition constituted by symmetric operators between 2 integrand objects, asymmetric operators between integrand and real numbers (and viceversa) and an assignment statement for the creation of new integrand objects. These procedures are analyzed in the following paragraphs.

### 3.1.1. Time derivative procedure, the residuals function

The abstract interface of the time derivative procedure *t* is shown in listing 3.

```
function time_derivative(self , t) result(dState_dt)
import :: integrand , R_P , I_P
class(integrand), intent(IN) :: self           !< Integrand field.
real(R_P),    optional, intent(IN) :: t          !< Time.
class(integrand), allocatable      :: dState_dt !< Result of the time derivative function of integrand field.
endfunction time_derivative
```

Listing 3: time derivative procedure interface

This procedure-function takes two arguments, the first passed as a *type bounded* argument, while the latter is optional, and it returns an integrand object. The passed dummy argument, *self*, is a polymorphic argument that could be any extensions of the *integrand* ADT. The optional argument *t* is the *time* at which the residuals function must be computed: it can be omitted in the case the residuals function does not depend directly on time.

Commonly, into the concrete implementation of this deferred abstract procedure clients embed the actual ODE equations being solved. As an example, for the Burgers equation, that is a Partial Differential Equations (PDE) system involving also a boundary value problem, this procedure embeds the spatial operator that convert the PDE to a system of algebraic ODE. As a consequence, the eventual concrete implementation of this procedure can be very complex and errors-prone. Nevertheless, the FOODiE solvers are implemented only on the above abstract interface, thus emancipating the solvers implementation from any concrete complexity.

Add citations to Burgers, Adams-Bashfort, leapfrog references.

### 3.1.2. Symmetric operators procedures

The abstract interface of *symmetric* procedures is shown in listing 4.

```
function symmetric_operator(lhs , rhs) result(operator_result)
import :: integrand
class(integrand), intent(IN) :: lhs           !< Left hand side.
class(integrand), intent(IN) :: rhs           !< Right hand side.
class(integrand), allocatable :: operator_result !< Operator result.
endfunction symmetric_operator
```

Listing 4: symmetric operator procedure interface

This interface defines a class of procedures operating on 2 *integrand* objects, namely it is used for the definition of the operators *multiplication*, *summation* and *subtraction* of integrand objects. These operators are used into the above described ODE solvers, for example see equations 2, 3, 6 or 9. The implementation details of such a procedures class are strictly dependent on the concrete extension of the integrand type. From the FOODiE solvers point of view, we need to known only that first argument passed as bounded one, the left-hand-side of the operator, and the second argument, the right-hand-side of the operator, are two integrand object and the returned object is still an integrand one.

### 3.1.3. Integrand/real and real/integrand operators procedures

The abstract interfaces of *Integrand/real and real/integrand operators* procedures are shown in listing 5.

```
function integrand_op_real(lhs , rhs) result(operator_result)
import :: integrand , R_P
class(integrand), intent(IN) :: lhs           !< Left hand side.
real(R_P),    intent(IN) :: rhs           !< Right hand side.
class(integrand), allocatable :: operator_result !< Operator result.
endfunction integrand_op_real
```

```

function real_op_integrand(lhs , rhs) result(operator_result)
import :: integrand , R_P
real(R_P), intent(IN) :: lhs           !< Left hand side.
class(integrand), intent(IN) :: rhs       !< Right hand side.
class(integrand), allocatable :: operator_result !< Operator result.
endfunction real_op_integrand

```

Listing 5: Integrand/real and real/integrand operators procedure interfaces

These two interfaces are necessary in order to complete the *algebra* operating on the integrand object class, allowing the multiplication of an integrand object for a real number, circumstance that happens in all solvers, see equations 2, 3, 6 or 9. The implementation details of these procedures are strictly dependent on the concrete extension of the integrand type. From the FOODiE solvers point of view, we need to know only that first argument passed as bounded one, the left-hand-side of the operator, and the second argument, the right-hand-side of the operator, are an integrand object and real number of viceversa and the returned object is still an integrand one.

### 3.1.4. Integrand assignment procedure

The abstract interface of *integrand assignment* procedure is shown in listing 6.

```

subroutine assignment_integrand(lhs , rhs)
import :: integrand
class(integrand), intent(INOUT) :: lhs !< Left hand side.
class(integrand), intent(IN)      :: rhs !< Right hand side.
endsubroutine assignment_integrand

```

Listing 6: integrand assignment procedure interface

The assignment statement is necessary in order to complete the *algebra* operating on the integrand object class, allowing the assignment of an integrand object by another one, circumstance that happens in all solvers, see equations 2, 3, 6 or 9. The implementation details of this assignment is strictly dependent on the concrete extension of the integrand type. From the FOODiE solvers point of view, we need to know only that first argument passed as bounded one, the left-hand-side of the assignment, and the second argument, the right-hand-side of the assignment, are two integrand objects.

### 3.2. The explicit forward Euler solver

The explicit forward Euler solver is exposed (by the FOODiE main module that must be imported, see listing 1) as a single derived type (that is a standard convention for all FOODiE solvers) named *euler\_explicit\_integrator*. It provides the type bound procedure (also referred as *method*) *integrate* for integrating in time an *integrand* object, or any of its polymorphic concrete extensions. Consequently, for using such a solver it must be previously defined as an instance of the exposed FOODiE integrator type, see listing 7.

```
use FOODiE, only: euler_explicit_integrator
type(euler_explicit_integrator) :: integrator
```

Listing 7: definition of an explicit forward Euler integrator

Once an integrator of this type has been instantiated, it can be directly used without any initialization, for example see listing 8.

```
type(my_integrand) :: my_field
call integrator%integrate(U=my_field, Dt=0.1)
```

Listing 8: example of usage of an explicit forward Euler integrator

where *my\_integrand* is a concrete (valid) extension of *integrand* ADT.

The complete implementation of the *integrate* method of the explicit forward Euler solver is reported in listing 9.

```
subroutine integrate(U, Dt, t)
class(integrand), intent(INOUT) :: U !< Field to be integrated.
real(R_P), intent(IN) :: Dt !< Time step.
real(R_P), optional, intent(IN) :: t !< Time.
U = U + U%t(t=t) * Dt
return
endsubroutine integrate
```

Listing 9: implementation of the *integrate* method of Euler solver

This method takes three arguments, the first argument is an integrand class, it being the integrand field that must be integrated one-step-over in time, the second is the time step used and the third, that is optional, is the current time value that is passed to the residuals function for taking into account the cases where the time derivative explicitly depends on time. The time step is not automatically computed (for example inspecting the passed integrand field), thus its value must be externally computed and passed to the *integrate* method.

### 3.3. The explicit TVD/SSP Runge-Kutta class of solvers

The TVD/SSP Runge-Kutta class of solvers is exposed as a single derived type named *tvd\_runge\_kutta\_integrator*. This type provides three methods:

- *init*: initialize the integrator accordingly the possibilities offered by the class of solvers;
- *destroy*: destroy the integrator previously initialized, eventually freeing the allocated dynamic memory registers;
- *integrate*: integrate integrand field one-step-over in time.

As common for FOODiE solvers, for using such a solver it must be previously defined as an instance of the exposed FOODiE integrator type, see listing 10.

```
use FOODiE, only: tvd_runge_kutta_integrator
type(tvd_runge_kutta_integrator) :: integrator
```

Listing 10: definition of an explicit TVD/SSP Runge-Kutta integrator

Once an integrator of this type has been instantiated, it must be initialized before used, for example see listing 11.

```
call integrator%init(stages=3)
```

Listing 11: example of initialization of an explicit TVD/SSP Runge-Kutta integrator

In the listing 11 a 3-stages solver has been initialized. As a matter of facts, from the equations 3 and 4 a solver belonging to this class is completely defined once the number of stages adopted has been chosen. The complete definition of the *tvd\_runge\_kutta\_integrator* type is reported into listing 12. As shown, the Butcher's coefficients are stored as allocatable arrays the values of which are initialized by the *init* method.

```
type :: tvd_runge_kutta_integrator
  integer(I_P) :: stages=0 ! Number of stages.
  real(R_P), allocatable :: alph(:, :) ! alpha Butcher's coefficients.
  real(R_P), allocatable :: beta(:) ! beta Butcher's coefficients.
  real(R_P), allocatable :: gamm(:) ! gamma Butcher's coefficients.
contains
  procedure, pass(self), public :: destroy
  procedure, pass(self), public :: init
  procedure, pass(self), public :: integrate
  final :: finalize
endtype tvd_runge_kutta_integrator
```

Listing 12: definition of *tvd\_runge\_kutta\_integrator* type

After the solver has been initialized it can be used for integrating an integrand field, as shown in listing 13.

```
type(my_integrand) :: my_field
type(my_integrand) :: my_stages(1:3)
call integrator%integrate(U=my_field, stage=my_stage, Dt=0.1)
```

Listing 13: example of usage of a TVD/SSP Runge-Kutta integrator

where *my\_integrand* is a concrete (valid) extension of *integrand* ADT. Listing 13 shows that the memory registers necessary for storing the Runge-Kutta stages must be supplied by the client code.

The complete implementation of the *integrate* method of the explicit TVD/SSP Runge-Kutta class of solvers is reported in listing 14.

```
subroutine integrate(self, U, stage, Dt, t)
  class(tvd_runge_kutta_integrator), intent(IN) :: self ! Actual RK integrator.
  class(integrand), intent(INOUT) :: U ! Field to be integrated.
  class(integrand), intent(INOUT) :: stage(1:) ! Runge-Kutta stages [1:stages].
  real(R_P), intent(IN) :: Dt ! Time step.
  real(R_P),
  integer(I_P) :: s ! First stages counter.
  integer(I_P) :: ss ! Second stages counter.
select type(stage)
class is(integrand)
  do s=1, self%stages
    stage(s) = U
    do ss=1, s - 1
      stage(s) = stage(s) + stage(ss) * (Dt * self%alph(s, ss))
    enddo
    stage(s) = stage(s)%t(t=t + self%gamm(s) * Dt)
  enddo
  do s=1, self%stages
    U = U + stage(s) * (Dt * self%beta(s))
  enddo
endselect
return
endsubroutine integrate
```

Listing 14: implementation of the *integrate* method of explicit TVD/SSP Runge-Kutta class

This method takes five arguments, the first argument is passed as bounded argument and it is the solver itself, the second is of an integrand class, it being the integrand field that must integrated one-step-over in time, the third is the stages array for storing the stages computations, the fourth is the time step used and the fifth, that is optional, is the current time value that is passed to the residuals function for taking into account the cases where the time derivative explicitly depends on time. The time step is not automatically computed (for example inspecting the passed integrand field), thus its value must be externally computed and passed to the *integrate* method.

It is worth noting that the stages memory registers, namely the array *stage*, must be passed as argument because it is defined as a *not-passed* polymorphic argument, thus we are not allowed to define it as an automatic array of the *integrate* method.

### 3.4. The explicit low storage Runge-Kutta class of solvers

The low storage variant of Runge-Kutta class of solvers is exposed as a single derived type named *ls\_runge\_kutta\_integrator*. This type provides three methods:

- *init*: initialize the integrator accordingly the possibilities offered by the class of solvers;
- *destroy*: destroy the integrator previously initialized, eventually freeing the allocated dynamic memory registers;
- *integrate*: integrate integrand field one-step-over in time.

As common for FOODiE solvers, for using such a solver it must be previously defined as an instance of the exposed FOODiE integrator type, see listing 15.

```
use FOODiE, only: ls_runge_kutta_integrator
type(ls_runge_kutta_integrator) :: integrator
```

Listing 15: definition of an explicit low storage Runge-Kutta integrator

Once an integrator of this type has been instantiated, it must be initialized before used, for example see listing 16.

```
call integrator%init (stages=5)
```

Listing 16: example of initialization of an explicit low storage Runge-Kutta integrator

In the listing 16 a 5-stages solver has been initialized. As a matter of facts, from the equation 5 a solver belonging to this class is completely defined once the number of stages adopted has been chosen. The complete definition of the *ls\_runge\_kutta\_integrator* type is reported into listing 17. As shown, the Williamson's coefficients are stored as allocatable arrays the values of which are initialized by the *init* method.

```
type :: ls_runge_kutta_integrator
  integer(I_P) :: stages=0 ! Number of stages.
  real(R_P), allocatable :: A(:) ! Low storage *A* coefficients.
  real(R_P), allocatable :: B(:) ! Low storage *B* coefficients.
  real(R_P), allocatable :: C(:) ! Low storage *C* coefficients.
contains
  procedure, pass(self), public :: destroy
  procedure, pass(self), public :: init
  procedure, pass(self), public :: integrate
  final :: finalize
endtype ls_runge_kutta_integrator
```

Listing 17: definition of *ls\_runge\_kutta\_integrator* type

After the solver has been initialized it can be used for integrating an integrand field, as shown in listing 18.

```
type(my_integrand) :: my_field
type(my_integrand) :: my_stages(1:2)
call integrator%integrate(U=my_field, stage=my_stage, Dt=0.1)
```

Listing 18: example of usage of a low storage Runge-Kutta integrator

where *my\_integrand* is a concrete (valid) extension of *integrand* ADT. Listing 18 shows that the memory registers necessary for storing the Runge-Kutta stages must be supplied by the client code, as it happens of the TVD/SSP Runge-Kutta class. However, now the registers necessary is always 2, independently on the number of stages used, that in the example considered are 5.

The complete implementation of the *integrate* method of the explicit low storage Runge-Kutta class of solvers is reported in listing 19.

```

subroutine integrate(self, U, stage, Dt, t)
  class(ls_runge_kutta_integrator), intent(IN)    :: self      ! Actual RK integrator.
  class(integrand),           intent(INOUT)   :: U        ! Field to be integrated.
  class(integrand),           intent(INOUT)   :: stage(1:2) ! Runge-Kutta registers [1:2].
  real(R_P),                  intent(IN)       :: Dt       ! Time step.
  real(R_P),                  intent(IN)       :: t        ! Time.
  integer(I_P)                :: s        ! First stages counter.

  select type(stage)
  class is(integrand)
    stage(1) = U
    stage(2) = U*0._R_P
    do s=1, self%stages
      stage(2) = stage(2) * self%A(s) + stage(1)%t(t=t + self%C(s) * Dt) * Dt
      stage(1) = stage(1) + stage(2) * self%B(s)
    enddo
    U = stage(1)
  endselect
  return
endsubroutine integrate

```

Listing 19: implementation of the *integrate* method of explicit low storage Runge-Kutta class

This method takes five arguments, the first argument is passed as bounded argument and it is the solver itself, the second is of an integrand class, it being the integrand field that must integrated one-step-over in time, the third is the stages array for storing the stages computations, the fourth is the time step used and the fifth, that is optional, is the current time value that is passed to the residuals function for taking into account the cases where the time derivative explicitly depends on time. The time step is not automatically computed (for example inspecting the passed integrand field), thus its value must be externally computed and passed to the *integrate* method.

It is worth noting that the stages memory registers, namely the array *stage*, must be passed as argument because it is defined as a *not-passed* polymorphic argument, thus we are not allowed to define it as an automatic array of the *integrate* method.

### 3.5. The explicit Adams-Bashforth class of solvers

The explicit Adams-Bashforth class of solvers is exposed as a single derived type named *adams\_bashforth\_integrator*. This type provides three methods:

- *init*: initialize the integrator accordingly the possibilities offered by the class of solvers;
- *destroy*: destroy the integrator previously initialized, eventually freeing the allocated dynamic memory registers;
- *integrate*: integrate integrand field one-step-over in time;
- *update\_previous*: auto update (cyclically) previous time steps solutions.

As common for FOODiE solvers, for using such a solver it must be previously defined as an instance of the exposed FOODiE integrator type, see listing 20.

```

use FOODiE, only: adams_bashforth_integrator
type(adams_bashforth_integrator) :: integrator

```

Listing 20: definition of an explicit Adams-Bashforth integrator

Once an integrator of this type has been instantiated, it must be initialized before used, for example see listing 21.

```

call integrator%init(steps=4)

```

Listing 21: example of initialization of an explicit Adams-Bashforth integrator

In the listing 21 a 4-steps solver has been initialized. As a matter of facts, from the equation 6 a solver belonging to this class is completely defined once the number of time steps adopted has been chosen. The complete definition of the *adams\_bashforth\_integrator* type is reported into listing 22. As shown, the linear coefficients are stored as allocatable arrays the values of which are initialized by the *init* method.

```

type :: adams_bashforth_integrator
private
integer(I_P) :: steps=0 ! Number of time steps.
real(R_P), allocatable :: b(:) ! b coefficients.
contains
procedure, pass(self), public :: destroy
procedure, pass(self), public :: init
procedure, pass(self), public :: integrate
procedure, pass(self), public :: update_previous
final :: finalize
endtype adams_bashforth_integrator

```

Listing 22: definition of *adams\_bashforth\_integrator* type

After the solver has been initialized it can be used for integrating an integrand field, as shown in listing 23.

```

real :: times(1:4)
type(my_integrand) :: my_field
type(my_integrand) :: previous(1:4)
call integrator%integrate(U=my_field, previous=previous, Dt=Dt, t=times)

```

Listing 23: example of usage of an Adams-Bashforth integrator

where *my\_integrand* is a concrete (valid) extension of *integrand* ADT, *times* are the time at each 4 steps considered for the current one-step-over integration and *previous* are the memory registers where previous time steps solutions are saved.

The complete implementation of the *integrate* method of the explicit Adams-Bashforth class of solvers is reported in listing 24.

```

subroutine integrate(self, U, previous, Dt, t, autoupdate)
class(adams_bashforth_integrator), intent(IN) :: self ! Actual AB integrator.
class(integrand), intent(INOUT) :: U ! Field to be integrated.
class(integrand), intent(INOUT) :: previous(1:) ! Previous time steps solutions.
real(R_P), intent(IN) :: Dt ! Time steps.
real(R_P), intent(IN) :: t(:) ! Times.
logical, optional, intent(IN) :: autoupdate ! Autoupdate previous time steps.
logical, intent(IN) :: autoupdate_ ! autoupdate previous time steps, dummy var.
integer(I_P) :: s ! Steps counter.
autoupdate_ = .true. ; if (present(autoupdate)) autoupdate_ = autoupdate
do s=1, self%steps
    U = U + previous(s)%t(t=t(s)) * (Dt * self%b(s))
enddo
if (autoupdate_) call self%update_previous(U=U, previous=previous)
return
endsubroutine integrate

```

Listing 24: implementation of the *integrate* method of explicit Adams-Bashforth class

This method takes five arguments, the first argument is passed as bounded argument and it is the solver itself, the second is of an integrand class, it being the integrand field that must integrated one-step-over in time, the third are the previous time steps solutions, the fourth is the time step used, the fifth is an array of the time values of the steps considered for the current one-step-over integration that are passed to the residuals function for taking into account the cases where the time derivative explicitly depends on time and the sixth is a logical flag for enabling/disabling the cyclic update of previous time steps solutions. The time step is not automatically computed (for example inspecting the passed integrand field), thus its value must be externally computed and passed to the *integrate* method.

It is worth noting that the method also performs the cyclic update of the previous time steps solutions memory registers. This can be disable passing *autoupdate=false*: it is useful in the framework of predictor-corrector solvers.

### 3.6. The implicit Adams-Moulton class of solvers

The explicit Adams-Moulton class of solvers is exposed as a single derived type named *adams\_moulton\_integrator*. This type provides three methods:

- *init*: initialize the integrator accordingly the possibilities offered by the class of solvers;
- *destroy*: destroy the integrator previously initialized, eventually freeing the allocated dynamic memory registers;
- *integrate*: integrate integrand field one-step-over in time;
- *update\_previous*: auto update (cyclically) previous time steps solutions.

As common for FOODiE solvers, for using such a solver it must be previously defined as an instance of the exposed FOODiE integrator type, see listing 25.

```
use FOODiE, only: adams_moulton_integrator
type(adams_moulton_integrator) :: integrator
```

Listing 25: definition of an implicit Adams-Moulton integrator

Once an integrator of this type has been instantiated, it must be initialized before used, for example see listing 26.

```
call integrator%init(steps=3)
```

Listing 26: example of initialization of an implicit Adams-Moulton integrator

In the listing 26 a 3-steps solver has been initialized. As a matter of facts, from the equation 7 a solver belonging to this class is completely defined once the number of time steps adopted has been chosen. The complete definition of the *adams\_moulton\_integrator* type is reported into listing 27. As shown, the linear coefficients are stored as allocatable arrays the values of which are initialized by the *init* method.

```
type :: adams_moulton_integrator
private
integer(I_P) :: steps=-1 ! Number of time steps.
real(R_P), allocatable :: b(:) ! b coefficients.
contains
procedure, pass(self), public :: destroy
procedure, pass(self), public :: init
procedure, pass(self), public :: integrate
procedure, pass(self), public :: update_previous
final :: finalize
endtype adams_moulton_integrator
```

Listing 27: definition of *adams\_moulton\_integrator* type

After the solver has been initialized it can be used for integrating an integrand field, as shown in listing 28.

```
real :: times(1:3)
type(my_integrand) :: my_field
type(my_integrand) :: previous(1:3)
call integrator%integrate(U=my_field, previous=previous, Dt=Dt, t=times)
```

Listing 28: example of usage of an Adams-Moulton integrator

where *my\_integrand* is a concrete (valid) extension of *integrand* ADT, *times* are the time at each 4 steps considered for the current one-step-over integration and *previous* are the memory registers where previous time steps solutions are saved.

The complete implementation of the *integrate* method of the implicit Adams-Moulton class of solvers is reported in listing 29.

```
subroutine integrate(self, U, previous, Dt, t, autoupdate)
class(adams_bashforth_integrator), intent(IN) :: self ! Actual AB integrator.
class(integrand), intent(INOUT) :: U ! Field to be integrated.
class(integrand), intent(INOUT) :: previous(1:) ! Previous time steps solutions.
real(R_P), intent(IN) :: Dt ! Time steps.
real(R_P), intent(IN) :: t(:) ! Times.
logical, optional, intent(IN) :: autoupdate ! Autoupdate previous time steps.
logical, intent(IN) :: autoupdate_ ! autoupdate previous time steps, dummy var.
integer(I_P) :: s ! Steps counter.
autoupdate_ = .true. ; if (present(autoupdate)) autoupdate_ = autoupdate
```

```

if (autoupdate_) call self%update_previous(U=U, previous=previous)
if (self%steps >0) then
    U = previous(self%steps) + U%t(t=t(self%steps) + Dt) * (Dt * self%b(self%steps))
    do s=0, self%steps - 1
        U = U + previous(s+1)%t(t=t(s+1)) * (Dt * self%b(s))
    enddo
    if (autoupdate_) call self%update_previous(U=U, previous=previous)
else
    U = U + U%t(t=t(s+1)) * (Dt * self%b(0))
endif
return
endsubroutine integrate

```

Listing 29: implementation of the *integrate* method of explicit Adams-Moulton class

This method takes six arguments, the first argument is passed as bounded argument and it is the solver itself, the second is of an integrand class, it being the integrand field that must integrated one-step-over in time, the third are the previous time steps solutions, the fourth is the time step used, the fifth is an array of the time values of the steps considered for the current one-step-over integration that are passed to the residuals function for taking into account the cases where the time derivative explicitly depends on time and the sixth is a logical flag for enabling/disabling the cyclic update of previous time steps solutions. The time step is not automatically computed (for example inspecting the passed integrand field), thus its value must be externally computed and passed to the *integrate* method.

It is worth noting that the method also performs the cyclic update of the previous time steps solutions memory registers. This can be disable passing *autoupdate=false*: it is useful in the framework of predictor-corrector solvers.

### 3.7. The predictor-corrector Adams-Bashforth-Moulton class of solvers

To be written.

### 3.8. The leapfrog solver

The explicit Leapfrog class of solvers is exposed as a single derived type named *leapfrog\_integrator*. This type provides three methods:

- *init*: initialize the integrator accordingly the possibilities offered by the class of solvers;
- *integrate*: integrate integrand field one-step-over in time.

As common for FOODiE solvers, for using such a solver it must be previously defined as an instance of the exposed FOODiE integrator type, see listing 30.

```

use FOODiE, only: leapfrog_integrator
type(leapfrog_integrator) :: integrator

```

Listing 30: definition of an explicit Leapfrog integrator

Once an integrator of this type has been instantiated, it must be initialized before used, for example see listing 31.

```

! default coefficients nu=0.01, alpha=0.53
call integrator%init()
! custom coefficients
call integrator%init(nu=0.015, alpha=0.6)

```

Listing 31: example of initialization of an explicit Leapfrog integrator

The complete definition of the *leapfrog\_integrator* type is reported into listing 32. As shown, the filter coefficients are initialized to zero, suitable values are initialized by the *init* method.

```

type :: leapfrog_integrator
private
    real(R_P) :: nu=0.01_R_P      ! Robert-Asselin filter coefficient.
    real(R_P) :: alpha=0.53_R_P   ! Robert-Asselin-Williams filter coefficient.
contains
    procedure , pass(self), public :: init

```

```

procedure , pass(self), public :: integrate
endtype leapfrog_integrator

```

Listing 32: definition of *leapfrog\_integrator* type

After the solver has been initialized it can be used for integrating an integrand field, as shown in listing 33.

```

real :: times(1:2)
type(my_integrand) :: filter_displacement
type(my_integrand) :: my_field
type(my_integrand) :: previous(1:2)
call integrator%integrate(U=my_field, previous=previous, filter=filter_displacement, Dt=Dt, &
                           t=times)

```

Listing 33: example of usage of a Leapfrog integrator

where *my\_integrand* is a concrete (valid) extension of *integrand* ADT, *previous* are the memory registers where previous time steps solutions are saved, *filter\_displacement* is the register necessary for computing the eventual displacement of the applied filter and *times* are the time at each 2 steps considered for the current one-step-over integration.

The complete implementation of the *integrate* method of the explicit Leapfrog class of solvers is reported in listing 34.

```

subroutine integrate(self, U, previous, Dt, t, filter)
class(leapfrog_integrator), intent(IN) :: self           ! LF integrator.
class(integrand),          intent(INOUT) :: U            ! Field to be integrated.
class(integrand),          intent(INOUT) :: previous(1:2) ! Previous time steps solutions.
real(R_P),                 intent(in)   :: Dt            ! Time step.
real(R_P),                 intent(IN)   :: t             ! Time.
class(integrand), optional, intent(INOUT) :: filter       ! Filter field displacement.
U = previous(1) + previous(2)%t(t=t) * (Dt * 2._R_P)
if (present(filter)) then
    filter = (previous(1) - previous(2) * 2._R_P + U) * self%nu * 0.5_R_P
    previous(2) = previous(2) + filter * self%alpha
    U = U + filter * (self%alpha - 1._R_P)
endif
previous(1) = previous(2)
previous(2) = U
return
endsubroutine integrate

```

Listing 34: implementation of the *integrate* method of explicit Leapfrog class

This method takes six arguments, the first argument is passed as bounded argument and it is the solver itself, the second is of an integrand class, it being the integrand field that must integrated one-step-over in time, the third are the previous time steps solutions, the fourth is the optional filter-displacement-register, the fifth is the time step used and the sixth is an array of the time values of the steps considered for the current one-step-over integration that are passed to the residuals function for taking into account the cases where the time derivative explicitly depends on time. The time step is not automatically computed (for example inspecting the passed integrand field), thus its value must be externally computed and passed to the *integrate* method. It is worth noting that if the filter displacement argument is not passed, the solver reverts back to the standard unfiltered Leapfrog method.

It is worth noting that the method also performs the cyclic update of the previous time steps solutions memory registers. In particular, if the filter displacement argument is passed the method performs the RAW filtering.

### 3.9. General Remarks

Table 10 presents a comparison of the relevant parts of equations 2, 3, 4, 5, 6 and 9 with the corresponding FOODiE implementations reported in listings 9, 14, 19, 24 and 34, respectively. This comparison proves that the *integrand* ADT has allowed a very high-level implementation syntax. The Fortran implementation is almost equivalent to the rigorous mathematical formulation. This aspect directly implies that the implementation of a ODE solver into the FOODiE library is very clear, concise and less-errors-prone than an *hard-coded* implementation where the solvers must be implemented for each specific definition of the integrand type, it being not abstract.

Table 10: Comparison between rigorous mathematical formulation and FOODiE high-level implementation; the syntax "(s)" and "(ss)" imply the summation operation

SOLVER	MATHEMATICAL FORMULATION	FOODiE IMPLEMENTATION
explicit forward Euler	$U(t^{n+1}) = U(t^n) + \Delta t \cdot R[t^n, U(t^n)]$	$U = U + U\%t(t = t) * Dt$
TVD/SSP Runge-Kutta	$K_s = R\left(t^n + \gamma_s \Delta t, U^n + \Delta t \sum_{l=1}^{s-1} \alpha_{s,l} K_l\right)$ $U^{n+1} = U^n + \Delta t \cdot \sum_{s=1}^{N_s} \beta_s K_s$	$stage(s) = stage(s) + stage(ss) * (Dt * self\%alph(s, ss))$ $U = U + stage(s) * (Dt * self\%beta(s))$
low storage Runge-Kutta	$K_2 = A_s K_2 + \Delta t \cdot R(t^n + C_s \Delta t, K_1)$ $K_1 = K_1 + B_s K_2$	$stage(2) = stage(2) * self\%A(s) +$ $+ stage(1)\%t(t = t + self\%C(s) * Dt) * Dt$ $stage(1) = stage(1) + stage(2) * self\%B(s)$
explicit Adams-Bashforth	$U(t^{n+N_s}) = U(t^{n+N_s-1}) +$ $+ \Delta t \sum_{s=1}^{n+N_s} b_s \cdot R\left[t^{n+s-1}, U(t^{n+s-1})\right]$	$U = U + U\%t(n = s, t = t(s)) * (Dt * self\%b(s))$
explicit Leapfrog	$U(t^{n+2}) = U(t^n) + 2\Delta t \cdot R\left[t^{n+1}, U(t^{n+1})\right]$	$U = U\%previous\_step(n = 1) + U\%t(n = 2, t = t) * (Dt * 2.)$

## 4. Tests and Examples

To be written.

### 4.1. Oscillation equations test

Let us consider the *oscillation* problem, it being a simple, yet interesting IVP. Briefly, the oscillation problem is a prototype problem of non dissipative, oscillatory phenomena. For example, let us consider a pendulum subjected to the Coriolis accelerations without dissipation, the motion equations of which can be described by the ODE system 11.

$$\begin{aligned} U_t &= R(U) \\ U = \begin{bmatrix} x \\ y \end{bmatrix} &R(U) = \begin{bmatrix} -fy \\ fx \end{bmatrix} \end{aligned} \quad (11)$$

where the frequency is chosen as  $f = 10^4$ . The ODE system 11 is completed by the following initial conditions:

$$\begin{aligned} x(t_0) &= 0 \\ y(t_0) &= 1 \end{aligned} \quad (12)$$

where  $t_0 = 0$  is the initial time considered.

The IVP constituted by equations 11 and 12 is (apparently) simple and its exact solution is known:

$$\begin{aligned} x(t_0 + \Delta t) &= X_0 \cos(f\Delta t) - y_0 \sin(f\Delta t) \\ y(t_0 + \Delta t) &= X_0 \sin(f\Delta t) + y_0 \cos(f\Delta t) \end{aligned} \quad (13)$$

where  $\Delta t$  is an arbitrary time step.

This problem is only apparently simple. As a matter of facts, in a non dissipative oscillatory problem the eventual errors in the amplitude approximation can rapidly drive the subsequent series of approximations to an unphysical solution. This is of particular relevance if the solution (that is numerically approximated) constitutes a *prediction* far from the initial conditions, that is the common case in weather forecasting.

#### 4.1.1. FOODiE aware implementation of an oscillation numerical solver

The IVP 11 can be easily solved by means of FOODiE library. The first block of a FOODiE aware solution consists to define an *oscillation integrand field* defining a concrete extension of the FOODiE *integrand* type. Listing 35 reports the implementation of such an integrand field that is contained into the tests suite shipped within the FOODiE library.

```
type , extends(integrand) :: oscillation
  private
    integer(I_P)                      :: dims=0      ! Space dimensions.
    real(R_P)                          :: f=0._R_P   ! Oscillation frequency (Hz).
    real(R_P), dimension(:), allocatable :: U          ! Integrand (state) variables, [1:dims].
  contains
    ! auxiliary methods
    procedure , pass(self), public :: init
    procedure , pass(self), public :: output
    ! type_integrand deferred methods
    procedure , pass(self), public :: t => dOscillation_dt
    procedure , pass(lhs),  public :: integrand_multiply_integrand => &
                                    oscillation_multiply_oscillation
    procedure , pass(lhs),  public :: integrand_multiply_real => oscillation_multiply_real
    procedure , pass(rhs),  public :: real_multiply_integrand => real_multiply_oscillation
    procedure , pass(lhs),  public :: add => add_oscillation
    procedure , pass(lhs),  public :: sub => sub_oscillation
    procedure , pass(lhs),  public :: assign_integrand => oscillation_assign_oscillation
    procedure , pass(lhs),  public :: assign_real => oscillation_assign_real
  endtype oscillation
```

Listing 35: implementation of the *oscillation integrand* type

The *oscillation* field extends the *integrand* ADT making it a concrete type. This derived type is very simple: it has 5 data members for storing the state vector and some auxiliary variables, and it implements all the deferred methods necessary for defining a valid concrete extension of the *integrand* ADT (plus 2 auxiliary methods that are not relevant for our discussion). The key point is here constituted by the implementation of the deferred methods: the *integrand* ADT does not impose any structure for the data members, that are consequently free to be customized by the client code. In this example the data members have a very simple, clean and concise structure:

- *dims* is the number of space dimensions; in the case of equation 11 we have *dims* = 2, however the test implementation has been kept more general parametrizing this dimension in order to easily allow future modification of the test-program itself;
- *f* stores the frequency of the oscillatory problem solved, that is here set to  $10^4$ , but it can be changed at runtime in the test-program;
- *U* is the state vector corresponding directly to the state vector of equation 11;

As the listing 35 shows, the FOODiE implementation strictly corresponds to the mathematical formulation embracing all the relevant mathematical aspects into one derived type, a single *object*. Here we not review the implementation of all deferred methods, this being out of the scope of the present work: the interested reader can see the tests suite sources shipped within the FOODiE library. However, some of these methods are relevant for our discussion, thus they are reviewed.

*dOscillation\_dt*, the *oscillation residuals* function. Probably, the most important methods for an IVP solver is the residuals function. As a matter of facts, the ODE equations are implemented into the residuals function. However, the FOODiE ADT strongly alleviates the subtle problems that could arise when the ODE solver is hard-implemented within the specific ODE equations. As a matter of facts, the *integrand* ADT specifies the precise interface the residuals function must have: if the client code implements a compliant interface, the FOODiE solvers will work as expected, reducing the possible errors location into the ODE equations, having designed the solvers on the ADT and not on the concrete type.

```
function dOscillation_dt(self, t) result(dState_dt)
  class(oscillation), intent(IN) :: self          ! Oscillation field.
  real(R_P), optional, intent(IN) :: t             ! Time.
  class(integrand), allocatable :: dState_dt      ! Oscillation field time derivative.
  integer(I_P) :: dn                            ! Time level, dummy variable.
  allocate(oscillation :: dState_dt)
  select type(dState_dt)
    class is(oscillation)
      dState_dt = self
      dState_dt%U(1) = -self%f * self%U(2)
      dState_dt%U(2) = self%f * self%U(1)
    endselect
  return
endfunction dOscillation_dt
```

Listing 36: implementation of the *oscillation integrand* residuals function

Listing 36 reports the implementation of the oscillation residuals function: it is very clear and concise. Moreover, comparing this listing with the equation 11 the close correspondence between the mathematical formulation and Fortran implementation is evident.

*add method, an example of oscillation symmetric operator*: As a prototype of the operators overloading let us consider the *add* operator, it being a prototype of symmetric operators, the implementation of which is presented in listing 37.

```
function add_oscillation(lhs, rhs) result(opr)
  class(oscillation), intent(IN) :: lhs  ! Left hand side.
  class(integrand), intent(IN) :: rhs  ! Right hand side.
  class(integrand), allocatable :: opr ! Operator result.
  allocate(oscillation :: opr)
  select type(opr)
```

```

class is( oscillation )
opr = lhs
select type(rhs)
class is ( oscillation )
opr%U = lhs%U + rhs%U
endselect
endselect
return
endfunction add_Oscillation

```

Listing 37: implementation of the *oscillation integrand* add operator

It is very simple and clear: firstly all the auxiliary data are copied into the operator result, then the state vector of the result is populated with the addiction between the state vectors of the left-hand-side and right-hand-side. This is very intuitive from the mathematical point of view and it helps to reduce implementation errors. Similar implementations are possible for all the other operators necessary to define a valid *intregrand* ADT concrete extension.

*assignment of an oscillation object.* The assignment overloading of the *oscillation* type is the last key-method that enforces the conciseness of the FOODiE aware implementation. Listing 38 reports the implementation of the assignment overloading. Essentially, to all the data members of the left-hand-side are assigned the values of the corresponding right-hand-side. Notably, for the assignment of the state vector and of the previous time steps solution array we take advantage of the automatic re-allocation of the left-hand-side variables when they are not allocated or allocated differently from the right-hand-side, that is a Fortran 2003 feature. In spite its simplicity, the assignment overloading is a key-method enabling the usage of FOODiE solver: effectively, the assignment between two *integrand* ADT variables is ubiquitous into the solvers implementations, see equation 3 for example.

```

subroutine oscillation_assign_oscillation(lhs , rhs)
class( oscillation ), intent(INOUT) :: lhs ! Left hand side.
class(integrand), intent(IN)      :: rhs ! Right hand side.
select type(rhs)
class is ( oscillation )
lhs%dims = rhs%dims
lhs%f = rhs%f
if (allocated(rhs%U)) lhs%U = rhs%U
endselect
return
endsubroutine oscillation_assign_oscillation

```

Listing 38: implementation of the *oscillation integrand* assignment

*FOODiE numerical integration.* Using the above discussed *oscillation* type it is very easy to solve IVP 11 by means of FOODiE library. Listing 39 presents the numerical integration of system 11 by means of the Leapfrog RAW-filtered method. In the example, the integration is performed with  $10^4$  steps with a fixed  $\Delta t = 10^2$  until the time  $t = 10^6$  is reached. The example shows also that for starting a multi-step scheme such as the Leapfrog one a lower-order or equivalent order one-scheme is necessary: in the example the first 2 steps are computed by means of one-step TVD/SSP Runge-Kutta 2-stages schemes. Note that the memory registers for storing the Runge-Kutta stages and the RAW filter displacement must be handled by the client code. Listing 39 demonstrates how it is simple, clear and concise to solve a IVP by FOODiE solvers. Moreover, it proves how it is simple and effective to apply different solvers in a coupled algorithm, that greatly simplify the development of new hybrid solvers for self-adaptive time step size.

```

use foodie , only: leapfrog_integrator , tvd_runge_kutta_integrator
type(leapfrog_integrator)      :: lf_integrator ! Leapfrog integrator.
type(tvd_runge_kutta_integrator) :: rk_integrator ! Runge-Kutta integrator.
type( oscillation )           :: rk_stage(1:2) ! Runge-Kutta stages.
type( oscillation )           :: previous(1:2) ! Previous time steps solution.
type( oscillation )           :: oscillator ! Oscillation field.
type( oscillation )           :: filter      ! Filter displacement.
integer                      :: step        ! Time steps counter.
real                          :: Dt          ! Time step.

```

```

call lf_integrator%init()
call rk_integrator%init(stages=2)
call oscillator%init(initial_state=[0.0, 1.0], f=10e4, steps=2)
Dt = 100.0
do step=1, 10000
  if (2>=step) then
    call rk_integrator%integrate(U=oscillator, stage=rk_stage, Dt=Dt, t=step*Dt)
    previous(step) = oscillator
  else
    call lf_integrator%integrate(U=oscillator, previous=previous, filter=filter, Dt=Dt, &
                                  t=step*Dt)
  endif
enddo
call print_results(U=oscillator)

```

Listing 39: numerical integration of the *oscillation* system by means of Leapfrog RAW-filtered method

#### 4.1.2. Errors Analysis of the FOODiE Oscillation test results

We use the Oscillation test program (shipped with FOODiE tests suite) for the testing the above described implementation and assessing the provided solvers accuracy for this IVP.

For the analysis of the accuracy of each solver, we have integrated the Oscillation equations 11 with different, decreasing time steps in the range [5000, 2500, 1250, 625, 320, 100]. The error is estimated by the L2 norm of the difference between the exact ( $U_e$ ) and the numerical ( $U_{\Delta t}$ ) solutions for each time step:

$$\varepsilon(\Delta t) = \|U_e - U_{\Delta t}\|_2 = \sqrt{\sum_{s=1}^{N_s} (U_e(t_0 + s * \Delta t) - U_{\Delta t}(t_0 + s * \Delta t))^2} \quad (14)$$

where  $N_s$  is the total number of time steps performed to reach the final integration time.

Using two pairs of subsequent-decreasing time steps solution is possible to estimate the order of accuracy of the solver employed computing the *observed order* of accuracy:

$$p = \frac{\log_{10}\left(\frac{\varepsilon(\Delta t_1)}{\varepsilon(\Delta t_2)}\right)}{\log_{10}\left(\frac{\Delta t_1}{\Delta t_2}\right)} \quad (15)$$

where  $\frac{\Delta t_1}{\Delta t_2} > 1$ .

Tables 11, 12, 13, 14 and 15 summarize the errors analysis.

Table 11: Oscillation test: errors analysis of explicit Adams-Bashforth solvers

(a) Adams-Bashforth 1 step					(b) Adams-Bashforth 2 steps				
TIME STEP	ERROR X	ERROR Y	ORDER X	ORDER Y	TIME STEP	ERROR X	ERROR Y	ORDER X	ORDER Y
5000.0	0.840E+10	0.706E+10	/	/	5000.0	0.596E+03	0.583E+03	/	/
2500.0	0.503E+06	0.570E+06	14.03	13.60	2500.0	0.221E+02	0.218E+02	4.75	4.74
1250.0	0.289E+04	0.272E+04	7.45	7.71	1250.0	0.764E+01	0.769E+01	1.53	1.50
625.0	0.239E+03	0.232E+03	3.59	3.55	625.0	0.265E+01	0.268E+01	1.53	1.52
320.0	0.737E+02	0.722E+02	1.76	1.74	320.0	0.968E+00	0.981E+00	1.51	1.50
100.0	0.250E+02	0.247E+02	0.93	0.92	100.0	0.169E+00	0.171E+00	1.50	1.50

(c) Adams-Bashforth 3 steps					(d) Adams-Bashforth 4 steps				
TIME STEP	ERROR X	ERROR Y	ORDER X	ORDER Y	TIME STEP	ERROR X	ERROR Y	ORDER X	ORDER Y
5000.0	0.857E+01	0.854E+01	/	/	5000.0	0.128E+07	0.143E+07	/	/
2500.0	0.391E+01	0.386E+01	1.13	1.14	2500.0	0.106E+01	0.107E+01	20.21	20.34
1250.0	0.825E+00	0.814E+00	2.24	2.25	1250.0	0.967E-01	0.981E-01	3.45	3.45
625.0	0.150E+00	0.148E+00	2.46	2.46	625.0	0.859E-02	0.871E-02	3.49	3.49
320.0	0.282E-01	0.278E-01	2.49	2.49	320.0	0.827E-03	0.838E-03	3.50	3.50
100.0	0.154E-02	0.152E-02	2.50	2.50	100.0	0.141E-04	0.143E-04	3.50	3.50

Table 12: Oscillation test: errors analysis of explicit Leapfrog solvers

(a) Leapfrog unfiltered					(b) Leapfrog RAW-filtered				
TIME STEP	ERROR X	ERROR Y	ORDER X	ORDER Y	TIME STEP	ERROR X	ERROR Y	ORDER X	ORDER Y
5000.0	0.156E+02	0.156E+02	/	/	5000.0	0.156E+02	0.156E+02	/	/
2500.0	0.849E+01	0.846E+01	0.87	0.88	2500.0	0.855E+01	0.852E+01	0.86	0.87
1250.0	0.300E+01	0.303E+01	1.50	1.48	1250.0	0.303E+01	0.305E+01	1.50	1.48
625.0	0.106E+01	0.107E+01	1.51	1.50	625.0	0.107E+01	0.108E+01	1.51	1.50
320.0	0.387E+00	0.392E+00	1.50	1.50	320.0	0.390E+00	0.395E+00	1.50	1.50
100.0	0.676E-01	0.685E-01	1.50	1.50	100.0	0.685E-01	0.692E-01	1.50	1.50

Table 13: Oscillation test: errors analysis of explicit Low Storage Runge-Kutta solvers

(a) Low storage Runge-Kutta 1 stage					(b) Low storage Runge-Kutta 5 stages				
TIME STEP	ERROR X	ERROR Y	ORDER X	ORDER Y	TIME STEP	ERROR X	ERROR Y	ORDER X	ORDER Y
5000.0	0.840E+10	0.706E+10	/	/	5000.0	0.120E+00	0.122E+00	/	/
2500.0	0.503E+06	0.570E+06	14.03	13.60	2500.0	0.106E-01	0.107E-01	3.51	3.51
1250.0	0.289E+04	0.272E+04	7.45	7.71	1250.0	0.935E-03	0.947E-03	3.50	3.50
625.0	0.239E+03	0.232E+03	3.59	3.55	625.0	0.826E-04	0.836E-04	3.50	3.50
320.0	0.737E+02	0.722E+02	1.76	1.74	320.0	0.793E-05	0.803E-05	3.50	3.50
100.0	0.250E+02	0.247E+02	0.93	0.92	100.0	0.135E-06	0.137E-06	3.50	3.50
(c) Low storage Runge-Kutta 6 stages					(d) Low storage Runge-Kutta 7 stages				
TIME STEP	ERROR X	ERROR Y	ORDER X	ORDER Y	TIME STEP	ERROR X	ERROR Y	ORDER X	ORDER Y
5000.0	0.979E-01	0.994E-01	/	/	5000.0	0.238E-01	0.240E-01	/	/
2500.0	0.876E-02	0.888E-02	3.48	3.48	2500.0	0.203E-02	0.205E-02	3.55	3.55
1250.0	0.776E-03	0.786E-03	3.50	3.50	1250.0	0.177E-03	0.180E-03	3.51	3.51
625.0	0.686E-04	0.695E-04	3.50	3.50	625.0	0.156E-04	0.158E-04	3.50	3.50
320.0	0.659E-05	0.667E-05	3.50	3.50	320.0	0.150E-05	0.152E-05	3.50	3.50
100.0	0.112E-06	0.114E-06	3.50	3.50	100.0	0.269E-07	0.273E-07	3.46	3.46

Table 14: Oscillation test: errors analysis of explicit TVD/SSP Runge-Kutta

(a) TVD/SSP Runge-Kutta 1 stage					(b) TVD/SSP Runge-Kutta 2 stages				
TIME STEP	ERROR X	ERROR Y	ORDER X	ORDER Y	TIME STEP	ERROR X	ERROR Y	ORDER X	ORDER Y
5000.0	0.840E+10	0.706E+10	/	/	5000.0	0.316E+02	0.319E+02	/	/
2500.0	0.503E+06	0.570E+06	14.03	13.60	2500.0	0.892E+01	0.894E+01	1.83	1.84
1250.0	0.289E+04	0.272E+04	7.45	7.71	1250.0	0.301E+01	0.305E+01	1.57	1.55
625.0	0.239E+03	0.232E+03	3.59	3.55	625.0	0.106E+01	0.107E+01	1.51	1.51
320.0	0.737E+02	0.722E+02	1.76	1.74	320.0	0.387E+00	0.392E+00	1.50	1.50
100.0	0.250E+02	0.247E+02	0.93	0.92	100.0	0.676E-01	0.685E-01	1.50	1.50
(c) TVD/SSP Runge-Kutta 3 stages					(d) TVD/SSP Runge-Kutta 5 stages				
TIME STEP	ERROR X	ERROR Y	ORDER X	ORDER Y	TIME STEP	ERROR X	ERROR Y	ORDER X	ORDER Y
5000.0	0.255E+01	0.252E+01	/	/	5000.0	0.139E+00	0.141E+00	/	/
2500.0	0.523E+00	0.516E+00	2.28	2.29	2500.0	0.122E-01	0.124E-01	3.50	3.50
1250.0	0.944E-01	0.931E-01	2.47	2.47	1250.0	0.108E-02	0.110E-02	3.50	3.50
625.0	0.167E-01	0.165E-01	2.50	2.50	625.0	0.956E-04	0.969E-04	3.50	3.50
320.0	0.314E-02	0.310E-02	2.50	2.50	320.0	0.937E-05	0.949E-05	3.47	3.47
100.0	0.171E-03	0.169E-03	2.50	2.50	100.0	0.512E-06	0.519E-06	2.50	2.50

*Adams-Basforth.* As expected, the Adams-Basforth 1 step solution, that reverts back to the explicit forward Euler one, is unstable for all the  $\Delta t$  exercised.

The expected observed orders of accuracy for the Adams-Basforth solvers using 2, 3 and 4 time steps tend to 1.5, 2.5 and 3.5 that are in agreement with the expected formal order of 2, 3 and 4, respectively.

Figure 1 emphasizes the instability generation for some pairs steps number/ $\Delta t$ . The 2 and 4 steps solutions are

Table 15: Oscillation test: errors analysis of predictor-corrector Adams-Bashforth-Moulton solvers

(a) Adams-Bashforth-Moulton 1 step					(b) Adams-Bashforth-Moulton 2 steps				
TIME STEP	ERROR X	ERROR Y	ORDER X	ORDER Y	TIME STEP	ERROR X	ERROR Y	ORDER X	ORDER Y
5000.0	0.241E+20	0.266E+20	/	/	5000.0	0.704E+01	0.701E+01	/	/
2500.0	0.664E+11	0.716E+11	28.44	28.47	2500.0	0.392E+01	0.395E+01	0.84	0.83
1250.0	0.952E+06	0.100E+07	16.09	16.12	1250.0	0.148E+01	0.150E+01	1.40	1.39
625.0	0.413E+04	0.407E+04	7.85	7.95	625.0	0.526E+00	0.534E+00	1.49	1.49
320.0	0.387E+03	0.383E+03	3.54	3.53	320.0	0.193E+00	0.196E+00	1.50	1.50
100.0	0.145E+03	0.145E+03	0.84	0.83	100.0	0.338E-01	0.342E-01	1.50	1.50
(c) Adams-Bashforth-Moulton 3 steps					(d) Adams-Bashforth-Moulton 4 steps				
TIME STEP	ERROR X	ERROR Y	ORDER X	ORDER Y	TIME STEP	ERROR X	ERROR Y	ORDER X	ORDER Y
5000.0	0.457E+01	0.464E+01	/	/	5000.0	0.229E+01	0.225E+01	/	/
2500.0	0.656E+00	0.654E+00	2.80	2.83	2500.0	0.119E+00	0.118E+00	4.26	4.25
1250.0	0.100E+00	0.987E-01	2.71	2.73	1250.0	0.825E-02	0.833E-02	3.85	3.83
625.0	0.169E-01	0.167E-01	2.56	2.56	625.0	0.671E-03	0.681E-03	3.62	3.61
320.0	0.314E-02	0.310E-02	2.52	2.51	320.0	0.631E-04	0.640E-04	3.53	3.53
100.0	0.171E-03	0.169E-03	2.50	2.50	100.0	0.107E-05	0.108E-05	3.51	3.51

instable for  $\Delta t = 5000 \rightarrow f * \Delta t = 0.5$ . On the contrary, the 3 steps solution is stable, but the amplitude is dumped and the solution vanishes as the integration proceeds. The 2 and 4 steps solutions show a phase error that decreases as the time resolution increases, whereas 3 steps solution has null phase error.

*Leapfrog.* The Leapfrog solutions are in agreement with the expected results: both unfiltered and RAW-filtered solutions show an observed order of accuracy that tends to the formal 2<sup>nd</sup> order. The two solutions are almost the same.

*Low Storage Runge-Kutta.*

*TVD/SSP Runge-Kutta.*




Figure 1: Oscillation equations solutions computed by means of Adams-Bashforth solvers




Figure 2: Oscillation equations solutions computed by means of Leapfrog RAW-filtered solver




Figure 3: Oscillation equations solutions computed by means of low storage Runge-Kutta solvers




Figure 4: Oscillation equations solutions computed by means of TVD/SSP Runge-Kutta solvers




Figure 5: Oscillation equations solutions computed by means of Adams-Bashforth-Moulton solvers

## **5. Benchmarks on parallel frameworks**

To be written.

### *5.1. OpenMP benchmark*

To be written.

### *5.2. MPI benchmark*

To be written.

## **6. Concluding Remarks and Perspectives**

To be written.

## **References**